USING THE SAP .NET CONNECTOR(Version 1.0.3)
Prepared by: Jay Kladiva version 1.0
COMMUNICATION MECHANISMS: RFC, BAPI, IDOC

CURRENT PROJECTS:  WMSC Wal-Mart Open Orders, BizTalk OTC-TMS submission of IDOCs to get transaction ID from SAP (TID)

USES:  Allows RFC and BAPI calls from .NET applications and can also be used to submit pre-generated IDOCs to SAP
ENVIRONMENTS:  Currently the SAP .NET Connector version 1.0.3 is installed on the USHQS055 Web Server, the USHQS093 BizTalk Server, and the USHQS092 BizTalk Server.

NOTES:  The SAP .NET Connector must be obtained from the download section of the SAP website.  BizTalk 2004 is only compatible with the 1.0.3 version of the SAP .NET Connector, other .NET projects can use newer versions of the SAP .NET Connector.  The SAP .NET Connector must be installed on the server that the .NET application will be running on.
IF YOU HAVE QUESTIONS:  Contact Jay Kladiva

STEPS TO CALL BAPI OR RFC:
Open Visual Studio.NET  2003 and under C# Projects, select SAP Connector, then select SAP server from the dialog box:

Enter the connection information for SAP and keep the defaults, you want to create a client proxy:

Enter part or all of the name of the RFC you wish to call, then select the RFC you are interested in and click the Add button:

The wizard will create a client proxy which will be utilized to call the RFC:

Rename this project with a suitable name and namespace and re-compile the project.  This assembly can now be called from one or more .NET projects.

Open your .NET project and set a reference to the assembly that was just created above and also set a reference to Energizer.SAP(this component can be pulled from Visual SourceSafe under the Components directory).

Implement a method which calls the RFC(see code example from WMSC Wal-Mart Open Orders project:
Public Function GetPOData(ByVal partnerNumber As String, ByVal PartnerFunction As String, ByVal orderNumbers() As String) As WMSC.SAP.Data.ZCUSTPO_DETTable

        ''' This Method is used to get PO Data from SAP. using the SAP Connector

        ''' 

        'throw an exception if the partner number is not 10 characters

        'should prefix the partner number using "0" (zero)

        If partnerNumber.Length <> 10 Then

            Throw New WMSCSAPException("Partner Number must be 10 characters")

        End If

        'Throw an exception if too many orders are passed in

        If orderNumbers.Length > MaxPONumbers Then

            Throw New WMSCSAPException("Can only pass up to " & MaxPONumbers & " order numbers per call.")

        End If

        'create an SAP Connection Object

        Dim sapDest As New Connector.Destination

        sapDest = InitializeConnection()

        Dim wmscProxy As New WMSC.SAP.Data.WMSCSAPProxy

        wmscProxy.Connection = New Connector.SAPConnection(sapDest)

        'create a New DETTable to store the order numbers

        Dim orders As New WMSC.SAP.Data.ZCUSTPO_DETTable

        'create a variable to be used to add orders into the orders DETTable 

        Dim order As WMSC.SAP.Data.ZCUSTPO_DET

        'loop through each order passed for our orderNumbers parameter

        For Each strOrder As String In orderNumbers

            'make sure the order number is not blank

            If Not Trim(strOrder) = "" Then

                'create a new instance of DET

                order = New WMSC.SAP.Data.ZCUSTPO_DET

                'Define the PO Number

                order.Cust_Po = strOrder

                'add to the orders table

                orders.Add(order)

            End If

        Next

        'create a new instance of DETTable to retain our results from the RFC

        Dim results As New WMSC.SAP.Data.ZCUSTPO_DETTable

        'create a string to pass in another required byref parameter for this RFC  

        Dim returncd As String

        'open the connection

        wmscProxy.Connection.Open()

        'call the function

        wmscProxy.Z_Cust_Po_Status("", partnerNumber, PartnerFunction, returncd, orders, results)

        'close the connection

        wmscProxy.Connection.Close()

        'return the results

        Return results

    End Function

STEPS TO SUBMIT IDOC:

The SAP .NET Connector can be utilized to submit IDOCs from any .NET application.  We currently utilize this in BizTalk to get a transaction ID from SAP(TID) from a given IDOC submission.
BizTalk Function call to custom method we created:

TID = SAPConnectorObj.SubmitIDOC(SAPHost, SAPClient, SAPSystemNumber, SAPUserName, SAPPassword, IDOCFilePath);

Custom class we implemented:

Public Class IDOCSubmit

    '// We call this method to submit the Idoc and return the TID value.  We need to pass in all the

    '// SAP communication information in order for this to occur.  The returned value is the string

    '// TID.

    Public Function SubmitIDOC(ByVal Host As String, ByVal Client As Int32, ByVal SystemNumber As Int32, _

        ByVal User As String, ByVal Password As String, ByVal IDOCPath As String) As String

        Dim tid As New SAP.Connector.RfcTID(System.Guid.NewGuid())

        Dim sapConnection As SAP.Connector.Destination

        Dim docSender As New SAP.Connector.SAPIDocSender

        Dim iCount As Int32 = 0

        Try

            '// we needed to write in this delay loop because we were frequently trying to 

            '// submit the Idoc before it was written to file by the send from the orchestration.

            Do While Not System.IO.File.Exists(IDOCPath)

                If iCount < 2 Then

                    System.Threading.Thread.CurrentThread.Sleep(1000)

                Else

                    Throw New ApplicationException("Unable to locate the IDOC file: " & IDOCPath)

                End If

                iCount += 1

            Loop

            sapConnection = InitializeConnection(Host, Client, SystemNumber, User, Password)

            docSender.Connection = New SAP.Connector.SAPConnection(sapConnection)

            docSender.SubmitIDoc(IDOCPath, tid)

            docSender.ConfirmTID(tid)

        Catch sapEx As SAP.Connector.RfcCommunicationException

            Throw New ApplicationException("SAP Communication Exception: " + sapEx.Message)

        Catch sapRfcEx As SAP.Connector.RfcException

            Throw New ApplicationException("SAP Rfc Exception: " + sapRfcEx.Message)

        Catch ioEx As IO.IOException

            '// this can happen if the BTS process has not released the file

            Dim msg As String = ioEx.Message

            Throw New ApplicationException("Error accessing file " + IDOCPath + ". Error message: " + msg, ioEx)

        Catch ex As Exception

            Throw New ApplicationException("Unexpected Error: " + ex.Message, ex)

        Finally

            If Not docSender Is Nothing Then docSender.Dispose()

            If Not sapConnection Is Nothing Then sapConnection.Dispose()

        End Try

        Return tid.ToString().Substring(0, 24)

    End Function

    '// This method creates a new Destination object which is used to establish the connection to

    '// the target SAP system.

    Private Function InitializeConnection(ByVal Host As String, ByVal Client As Int32, ByVal SystemNumber As Int32, _

        ByVal User As String, ByVal Password As String) As SAP.Connector.Destination

        Dim sapConnection As SAP.Connector.Destination

        ' initialize the connection

        sapConnection = New SAP.Connector.Destination

        sapConnection.AppServerHost = Host

        sapConnection.Client = Client

        sapConnection.Password = Password

        sapConnection.Username = User

        sapConnection.SystemNumber = SystemNumber

        Return sapConnection

    End Function

End Class

BizTalk Orchestration that utilizes the SAP .NET Connector:

